
Charu Shankar, SAS Inst it ute

With a background in computer systems management . SAS 
Inst ructor Charu Shankar engages with logic, visuals, and 
analogies to spark crit ical thinking since 20 0 7.

Charu curates and delivers unique content  on SAS, SQL,  Viya, 
etc. to support  users in the adopt ion of SAS software.

When not  coding,  Charu teaches yoga and loves to explore 
Canadian t rails with her husky Miko.

Confession 1: PROC SQL Syntax Order Mnemonic
Overview of the SQL Procedure

4

SQL Procedure
The SQL procedure is init iated with a PROC SQL statement . It  is  
terminated with a QUIT statement .

s102d01

PROC SQL <option(s)>;
statement(s);
QUIT;

proc sql;
select Employee_ID, Employee_Gender,
   Salary
   from mwsug.employee_information;
quit;

SQL Procedure

– Mult iple statements can be included in a PROC SQL step.
– Each statement  defines a process and is executed immediately.

PROC SQL <option(s)>;
statement(s);
QUIT;

SELECT Statement

A SELECT statement is  used to query one or more tables. The results of the 
SELECT statement  are written to the default  output  dest inat ion.

s102d01

proc sql;
select Employee_ID, Employee_Gender, Salary
   from mwsug.employee_information
   where Employee_Gender='F'
   order by Salary desc;
quit;

SELECT Statement
A SELECT statement  contains smaller building blocks called clauses. 

Note: Although it can contain multiple clauses, each  SELECT statement 
begins with the SELECT keyword and ends with a semicolon.

s102d01

clauses

proc sql;
select Employee_ID, Employee_Gender, Salary

from mwsug.employee_information
where Employee_Gender='F'
order by Salary desc;

quit;

Viewing the Output
Part ial PROC SQL Output

The SAS System

                                Employee
                Employee          Annual
   Employee ID  Gender            Salary
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
        120260  F               $207,885
        120719  F                $87,420
        120661  F                $85,495
        121144  F                $83,505
        120798  F                $80,755

SELECT Statement: Required Clauses

Here are two things that  SQL always needs:
1. What  do you want? 

The SELECT clause specifies the columns and column order.
2. Where do you want  it  from?

The FROM clause specifies the data sources. 
You can query from 1 to 256 tables.

SELECT object-item <, ...object-item>
 FROM from-list;

SELECT Statement: Syntax Order 
Mnemonic

– The WHERE clause specifies data that  meets certain condit ions.
– The GROUP BY clause groups data for processing.
– The HAVING clause specifies groups that  meet  certain condit ions.
– The ORDER BY clause specifies an order for the data.

SELECT object-item <, ...object-item>
       FROM from-list
 <WHERE sql-expression>
 <GROUP BY object-item <, … object-item >>
 <HAVING sql-expression>
 <ORDER BY order-by-item <DESC>
                              <, …order-by-item>>;

SO
FEW 
WORKERS
GO
HOME 
ON TIME

Discussion

Is this code correct?

proc sql;
select Employee_ID, Employee_Gender, Salary
   from mwsug.employee_information
   order by Employee_ID
   where Employee_Gender='M';
quit;

Syntax Check with the NOEXEC Option

To explicit ly check for syntax errors without  submitt ing  the code for execut ion, 
include the NOEXEC opt ion in the PROC SQL statement . This opt ion applies 
to all statements in a PROC SQL step. 

s102d03

PROC SQL <NOEXEC>;

proc sql noexec;
select Employee_ID, Employee_Gender, Salary
   from mwsug.employee_information
   where Employee_Gender=‘M'
   order by Salary desc;
quit;

Viewing the Log
Part ial SAS Log

proc sql noexec;
select Employee_ID, Employee_Gender, Salary
   from mwsug.employee_information
   where Employee_Gender=‘M'
   order by Salary desc;
NOTE: Statement not executed due to NOEXEC option.
quit;

Confession 2. Know Thy Data: 

Dict ionary Tables & Views

Business Scenario

You have inherited many different  data tables and want  to become familiar 
with their content .

mwsu
g

Sashelp

? ?

DICTIONARY Tables: Overview
• DICTIONARY tables are Read-Only metadata views that  contain 

session metadata, such as informat ion about  SAS libraries, data 
sets, and external files in use or available in the current  SAS 
session.

• DICTIONARY tables are
– created at  SAS session init ializat ion
– updated automat ically by SAS
– limited to Read-Only access.

• You can query DICTIONARY tables with PROC SQL. 

Querying Metadata about SAS Libraries

There can be more than 30  DICTIONARY tables. We will focus on 
using data from three of the tables.
 

• DICTIONARY.TABLES
 detailed information about tables
• DICTIONARY.COLUMNS
 detailed information about all columns in all tables
• DICTIONARY.MEMBERS

general information about SAS library members

Exploring DICTIONARY Tables

NOTE: SQL table DICTIONARY.TABLES was created like:

create table DICTIONARY.TABLES
  (
   libname char(8) label='Library Name',
   memname char(32) label='Member Name',
   ...  
   crdate num format=DATETIME informat=DATETIME label='Date Created',
   modate num format=DATETIME informat=DATETIME label='Date Modified',
   nobs num label='Number of Physical Observations',
   obslen num label='Observation Length',
   nvar num label='Number of Variables', ...);

s108d01

You can use a DESCRIBE statement  to explore the st ructure of DICTIONARY tables:

 Partial Log
describe table dictionary.tables;

Querying Dictionary Information
Display informat ion about  the tables in the SASHELP library.

s108d01

Library names are 
stored in uppercase 
in DICTIONARY tables.

title 'Tables in the SASHELP Library';
proc sql;
select memname 'Table Name',
       nobs,nvar,crdate
   from dictionary.tables
   where libname='SASHELP';
quit;

Viewing the Output
Part ial PROC SQL Output

Querying Dictionary Information
Display informat ion about  the columns in sashelp.cars.

s108d01

title 'Columns in the sashelp.cars Table';
proc sql;
select Name,Type,Length
   from dictionary.columns
   where libname='SASHELP'
         and memname='CARS';
quit;

Table names (memnames) 
are also stored in uppercase 
in DICTIONARY tables.

Viewing the Output
PROC SQL Output

Column names are stored in mixed case.

Using Dictionary Information

Which tables contain an ID column?

s108d01

Because different tables might use different cases
for same-named columns, you can use the UPCASE
function for comparisons. However, this significantly 
degrades the performance of the query.

title 'Tables Containing an ID Column';
proc sql;
select memname 'Table Names', name
   from dictionary.columns
   where libname='SASHELP' and 

upcase(name) contains 'ID';
quit;

Viewing the Output

All ID column names are stored in uniform 
uppercase, so the UPCASE function is not needed 
the next time that a query such as this is executed.

Finding Common Column Names Dynamically
All of the previous techniques to explore DICTIONARY tables work when you 
know the names of columns.
What  happens if you do not  know your data, and you want  SAS to ret rieve all 
same-named columns in a library.
Use the following code

s108d01

title 'Common columns in SASHELP';
proc sql;
select name, type, length, memname
   from dictionary.columns
   where libname='SASHELP’
   group by name

having count(name) > 1; 

Viewing the Output

Joins are easier because the st ructure of each table does not  have to be 
examined before determining common columns. Let  SAS bring common 
columns dynamically by looking up DICTIONARY tables.

Using DICTIONARY Tables in Other SAS Code
SAS provides views based on the DICTIONARY tables in the SASHELP library. 
Most of the SASHELP library DICTIONARY view names are similar to 
DICTIONARY table names, but they are shortened to eight characters or less. They 
begin with the letter v and do not end in s. For example:

dictionary.tables = sashelp.vtable
The following code executes successfully:

s108d01

title 'Tables in the SASHELP Library';
proc print data=sashelp.vtable NOOBS ;
   var memname nobs nvar;
   where libname='SASHELP';
run; 

options fullstimer;
proc sql;
 select libname, memname, name, type, length
 from dictionary.columns
 where upcase(name) contains 'ID'
 and libname='SASHELP' and type='num';
quit;
NOTE: PROCEDURE SQL used (Total process time):
      real time           0.73 seconds
      user cpu time       0.42 seconds
      system cpu time     0.29 seconds
      memory              5584.18k
      OS Memory           24672.00k
    

An Efficiency Question: PROC SQL or PRINT?

s108d01

Statistic Description

Real Time The amount of real time (clock time) spent to process the SAS job. Real time is also referred 
to as elapsed time.

User CPU Time The CPU time that is spent in the user program.

System CPU 
Time

CPU time is spent to perform operating system tasks (system overhead tasks) that support 
the execution of your SAS code.

Memory The amount of memory required to run a step.

OS Memory the largest amount of operating system memory that is available to SAS during the step.

Timestamp The date and time that a step was executed.

Step Count Count of DATA steps or procedures that run in a SAS program.

Switch Count A count of task switches within a step—that is, within a DATA step 
or procedure—in a SAS program. A task switch occurs when a step requests service from 
another process. Another task switch occurs when the step resumes. The number reported 
is for the last step that runs.

An Efficiency Question: PROC SQL or PRINT?
What  do these stat ist ics mean?

options fullstimer;
proc print data=sashelp.vcolumn;
 var libname memname name type length;
 where upcase(name) contains 'ID' and    
libname='SASHELP' and type='num';
run;

NOTE: There were 34 observations read from the data set SASHELP.VCOLUMN. 
WHERE UPCASE(name) contains 'ID' and (libname='SASHELP') and 
(type='num');
NOTE: PROCEDURE PRINT used (Total process time):
      real time           2.19 seconds
      user cpu time       0.92 seconds
      system cpu time     1.18 seconds
      memory              6738.81k
      OS Memory           25440.00k
    

An Efficiency Question: PROC SQL or PRINT?
Can I use PROC PRINT instead?

Confession 3. Stack Data Horizontally
Subqueries: Best  Pract ices, Dangers of Correlated

SELECT ...>
 FROM ...
 <WHERE ...>
         
 <GROUP BY ...>
 <HAVING ...>

 <ORDER BY ...>;

Subqueries

A subquery 
– returns values to be used in the outer query’s WHERE or HAVING clause

…(select Employee_ID
    from mwsug.staff
    where …  )…

...

proc sql;
select Job_Title, avg(Salary) as MeanSalary
   from mwsug.staff
   group by Job_Title
   having avg(Salary) >
      (select avg(Salary)
          from mwsug.staff);
quit;

Subqueries: Noncorrelated
There are two types of subqueries:
– A noncorrelated subquery is  a self-contained query. 

It  executes independent ly of the outer query.

This query is a stand-alone query.

Business Scenario
HR and Payroll managers requested a report  that  displays Job_Title for 
job groups with an average salary greater than the average salary of the 
company as a whole.

Employee Job Title       MeanSalary
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Account Manager               46090
Administration Manager        47415
Applications Developer I      42760
 ...

mwsug.staf
f     PROC SQL

Step 1

Calculate the company's average salary.

s105d01

Company
MeanSalary
ƒƒƒƒƒƒƒƒƒƒ
  38041.51

proc sql;
select avg(Salary) as CompanyMeanSalary
   from mwsug.staff;
quit;

Step 2
Determine the job t it les whose average salary exceeds the 
company’s average salary.

Part ial PROC SQL Output
Employee Job Title         MeanSalary
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Account Manager                 46090
Administration Manager          47415
Applications Developer I        42760 s105d01

proc sql;
select Job_Title, 
       avg(Salary) as MeanSalary
   from mwsug.staff
   group by Job_Title
   having MeanSalary>38041.51;
quit;

Step 3
• Write the program as a single step using a subquery. 
• A subquery is  a query that  resides within an outer query.

Note: The subquery must be resolved before the outer
 query can be resolved s105do1

proc sql;
select Job_Title, 
       avg(Salary) as MeanSalary
   from mwsug.staff
   group by Job_Title
   having MeanSalary >
  (select avg(salary) 
   from mwsug.staff);
quit;

outer query

subquery

Business Scenario

39

Each month, the CEO sends a birthday card to 
each employee having a birthday in that  month. 
Create a report  list ing the names and addresses 
of employees with February birthdays.

Noncorrelated Subqueries

s104d02

The mwsug.Employee_Addresses table contains names and 
addresses. Birth dates are found in the mwsug.Employee_Payroll 
table.

Write a query to grab the February born staff.

proc sql;
select Employee_ID
          from mwsug.Employee_Payroll
            where month(Birth_Date)=2;
quit;

Noncorrelated Subqueries

s104d02

Pass the query with the February borns as a subquery.
proc sql;
   select Employee_ID, 
          Employee_Name, City, 
          Country 
      from mwsug.Employee_Addresses
      where Employee_ID in
        (select Employee_ID

from mwsug.Employee_Payroll
where month(Birth_Date)=2)

      order by 1;
quit;

Noncorrelated Subqueries:
How Do They Work?

42

Partial

...s104d02

mwsug.Employee_Payroll
Employee

_ID
Birth
_Date

… …
120106 23DEC1948
120107 21JAN1953
120108 23FEB1988
120109 15DEC1990
120110 20NOV1953
120111 23JUL1953
120112 17FEB1973
120113 10MAY1948

… …

Step 1: Evaluate the inner query and
build a virtual table that satisfies the
WHERE criteria.

proc sql;
   select Employee_ID, 
          Employee_Name, City, 
          Country 
      from mwsug.Employee_Addresses
      where Employee_ID in
        (select Employee_ID

from mwsug.Employee_Payroll
where month(Birth_Date)=2)

      order by 1;
quit;

proc sql;
   select Employee_ID,           
          Employee_Name, City,  
          Country
      from mwsug.Employee_Addresses
      where Employee_ID in
        (120108,120112,120114,120157,

120159, 120170,…)
      order by 1;
quit;

Noncorrelated Subqueries:
How Do They Work?

43

...s104d02

Values returned by the inner query

Partial
mwsug.Employee_Payroll

Employee
_ID

Birth
_Date

… …

120106 23DEC1948
120107 21JAN1953
120108 23FEB1988
120109 15DEC1990
120110 20NOV1953
120111 23JUL1953
120112 17FEB1973
120113 10MAY1948

… …

proc sql;
   select Employee_ID,           
          Employee_Name, City,  
          Country
      from mwsug.Employee_Addresses
      where Employee_ID in
        (120108,120112,120114,120157,

120159, 120170,…)
      order by 1;
quit;

Noncorrelated Subqueries:
How Do They Work?

44

s104d02

mwsug.Employee_Payroll

Employee
_ID

Birth
_Date

… …

120106 23DEC1948

120107 21JAN1953

120108 23FEB1988

120109 15DEC1990

120110 20NOV1953

120111 23JUL1953

120112 17FEB1973

120113 10MAY1948

… …

Step 2: Pass the values to the outer
query for use in the WHERE clause.

Partial

Noncorrelated Subqueries: Output

45

The SAS System

Employee_ID   Employee_Name             City                Country
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
     120108   Gromek, Gladys             Melbourne  AU
     120112   Glattback, Ellis           Melbourne  AU
     120114   Buddery, Jeannette         Sydney  AU
     120157   Karavdic, Leonid           Sydney  AU
     120159   Phoumirath, Lynelle        Sydney  AU
     120170   Kingston, Alban            Sydney  AU

Do these look familiar?
They are the employee IDs
returned by the inner query.

Correlated Subqueries

• Correlated subqueries
– cannot  be evaluated independent ly
– require values to be passed to the inner query from the outer query
– are evaluated for each row in the outer query.

Subqueries: Correlated
A correlated subquery requires a value or values 
to be passed to it  by the outer (main) query before 
it  can be successfully resolved.

proc sql;
select Employee_ID, avg(Salary) as MeanSalary
   from mwsug.employee_addresses
   where 'AU'=
      (select Country
          from work.supervisors
          where employee_addresses.Employee_ID=
                supervisors.Employee_ID);
quit;

Business Scenario
Use a correlated subquery to create a report  list ing the employee 
ident ifier and name for all managers in Aust ralia.

Outer Query

PROC SQL

mwsug.employee_addresses

Inner Query

where employee_addresses.Employee_ID=supervisors.Employee_ID

work.supervisors 

Considerat ions:
 You have a temporary table, Supervisors, containing Employee_ID and Country 

for all managers.
 The table mwsug.Employee_Addresses contains Employee_Name for all 

employees

Correlated Subqueries
In a correlated subquery, the outer query provides 
informat ion so that  the subquery resolves successfully.

You must qualify each column with a table name.

s1aad01

proc sql;
  select Employee_ID,
         Employee_name
     from mwsug.Employee_Addresses
     where 'AU'=
       (select Country
        from Work.Supervisors
      where Employee_Addresses.Employee_ID=

Supervisors.Employee_ID);
quit;

This query is not stand-alone.
It needs additional information
from the main query.

Confession 4. Where ANSI falls short and PROC 
SQL steps in
Making a view portable

Business Scenario
Tom Zhou is a sales manager who needs access to personnel 
informat ion for his staff.

Tom Zhou

Business Data
The data that  Tom needs is name, job t it le, salary, 
and years of service. Data is contained in three tables.

                           mwsug.employee_addresses

                       mwsug.employee_payroll

                       mwsug.employee_organization

Considerations
• What  is the best  way to help Tom, given the following requirements:

– He should not  be allowed access to personnel data for any employee 
that  is not  his direct  report .

– He can write simple PROC SQL queries and use basic SAS 
procedures, but  cannot  write complex joins.

• A PROC SQL view accessing data for Tom Zhou’s direct  reports can 
provide the informat ion that  Tom needs in a secure manner.

What Is a PROC SQL View?
A PROC SQL view
– is a stored query
– contains no actual data
– can be derived from one or more tables, PROC SQL views, 

DATA step views, or SAS/ACCESS views
– extracts underlying data each t ime it  is  used and accesses 

the most  current  data
– can be referenced in SAS programs in the same way as a data 

table
– cannot  have the same name as a data table stored 

in the same SAS library.

Creating a PROC SQL View
To create a PROC SQL view, use the CREATE VIEW statement .

s107d07

CREATE VIEW view-name AS
        SELECT …;

proc sql;
create view mwsug.tom_zhou as
   select Employee_Name as Name format=$25.0,
          Job_Title as Title format=$15.0,
          Salary 'Annual Salary' format=comma10.2,
          int((today()-Employee_Hire_Date)/365.25) 
             as YOS 'Years of Service'
      from employee_addresses as a,
           employee_payroll as p,
           employee_organization as o
      where a.Employee_ID=p.Employee_ID and
            o.Employee_ID=p.Employee_ID and
            Manager_ID=120102;
quit;

View the Log
Part ial SAS Log
5      proc sql;
46  create view mwsug.tom_zhou as
47   select Employee_Name as Name format=$25.0,
48        Job_Title as Title format=$15.0,
49      Salary 'Annual Salary' format=comma10.2,
50              int((today()-Employee_Hire_Date)/365.25)
51                      as YOS 'Years of Service'
52               from employee_addresses as a,
53                    employee_payroll as p,
54                    employee_organization as o
55               where a.Employee_ID=p.Employee_ID and
56                     o.Employee_ID=p.Employee_ID and
57                     Manager_ID=120102;
NOTE: SQL view mwsug.TOM_ZHOU has been defined.

Location of a PROC SQL View
ANSI standards specify that  the view must  reside 
in the same SAS library as the contribut ing table or tables. 

libname=mwsug

Employee_Addresses

Tom_Zhou

s:\workshop

Employee_Organization

Employee_Payroll

Location of the Source Tables: ANSI 
In PROC SQL, the default  libref for the table (or tables) 
in the FROM clause is the libref of the library that  contains the view. 
When the view and data source are in the same locat ion, you specify a 
one-level name for the table (or tables) in the FROM clause. 

create view mwsug.tom_zhou as
   …
   from employee_addresses as a,
        employee_payroll as p,
        employee_organization as o

Using a View
Tom can use the view to produce simple reports.

Part ial PROC SQL Output  (executed 5June20 24)

s107d08

title "Tom Zhou's Direct Reports";
title2 "By Title and Years of 
Service";
select *

from mwsug.tom_zhou
   order by Title desc, YOS desc;

Business Scenario

• You created a PROC SQL view to provide Tom Zhou access to personnel data 
for his direct  reports.

• Tom copied his view to a folder on his hard drive. 
Now Tom reports that  the view does not  work anymore, and he asked for 
your help to resolve the problem.

Exploring the Problem

Tom submitted the following: 

s107d09

libname mwsug 'c:\temp';
proc sql;
title "Tom Zhou's Direct Reports";
title2 "By Title and Years of Service";
select *
   from mwsug.tom_zhou
   order by Title desc, YOS desc;
quit;
title;

Viewing the Log
Part ial SAS Log

libname mwsug ‘c:\workshop';
NOTE: Libref mwsug was successfully assigned as follows:
      Engine:        V9
      Physical Name: c:\workshop
   proc sql;
   title "Tom Zhou's Direct Reports";
   title2 "By Title and Years of Service";
   select *
      from mwsug.tom_zhou
      order by Title desc, YOS desc;
ERROR: File mwsug.EMPLOYEE_ADDRESSES.DATA does not exist.
ERROR: File mwsug.EMPLOYEE_PAYROLL.DATA does not exist.
ERROR: File mwsug.EMPLOYEE_ORGANIZATION.DATA does not exist.
   quit;
   title;
NOTE: The SAS System stopped processing this step because of errors.

A Violation
Tom moved his view to his C:\ workshop folder and redefined the mwsug library 
there. This violated the one-level naming convention in the FROM clause 
in the view code. 

libname mwsug ‘c:\workshop';
proc sql;
title "Tom Zhou's Direct Reports";
title2 "By Title and Years of Service";
select *
   from mwsug.tom_zhou
   order by Title desc, YOS desc;
quit;

Making a View Portable

s107d10

create view mwsug.Tom_Zhou as
   select Employee_Name as Name format=$25.0,
          Job_Title as Title format=$15.0,
          Salary "Annual Salary" format=comma10.2,
          int((today()-Employee_Hire_Date)/365.25)
             as YOS 'Years of Service'
      from mwsug.employee_addresses as a,
           mwsug.employee_payroll as p,
           mwsug.employee_organization as o
      where a.Employee_ID=p.Employee_ID and
            o.Employee_ID=p.Employee_ID and
            Manager_ID=120102
      using libname mwsug "s:\workshop";

CREATE VIEW view AS SELECT… 
        <USING LIBNAME-clause<, …LIBNAME-clause>>; 

A USING clause names the 
location of the tables.

two-level data 
set names

libname=mwsug

employee_addresses

s:\workshop

Two-Level Table Names in Permanent Views

– The USING clause libref is local to the view, 
and it  will not  conflict  with an ident ically named libref 
in the SAS session.

– When the query finishes, the libref is disassociated.

CREATE VIEW proc-sql-view AS SELECT …
        <USING LIBNAME-clause<, …LIBNAME-clause>>; 

libname=mwsug

tom_zhou

c:\workshop

employee_organization

employee_payroll

Views: Advantages
Advantages Disadvantages
• avoid storing copies of large tables.
• avoid a frequent refresh of table copies. 

When the underlying data changes, a 
view surfaces the most current data.

• pull together data from multiple 
database tables and multiple libraries or 
databases.

• simplify complex queries.
• prevent other users from inadvertently 

altering the query code.

• Because views access the most current 
data in changing tables, the results 
might be different each time you access 
the view.

• Views can require significant resources 
each time they execute. With a view, 
you save disk storage space at the cost 
of extra CPU and memory usage.

• When accessing the same data several 
times in a program, use a table instead 
of a view. This ensures consistent 
results from one step to the next and 
can significantly reduce the resources 
that are required.

Confession 5. Summarizing Data using the 
Boolean Gate
Summarizing Data

Business Scenario
Create a report  that  list s the following for each department :

– total number of managers 
– total number of non-manager employees
– manager-to-employee (M/ E) rat io

Below is a rough sketch of the desired report . 

Department Managers Employees
M/E 

Ratio
Accounts 1 5 20%
Administration 2 20 10%

Business Data
Determine whether an employee is a manager 
or a non-manager. 
The Job_Title column contains the information about each 
employee.  

Department        Job_Title                
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Administration    Administration 
Manager          
Administration    Secretary I                    
Administration    Office Assistant II            

Counting Rows That Meet a Specified Criterion

How do you determine the rows that  do have Manager 
in Job_Title, as well as rows that do not? You cannot use a WHERE clause to 
exclude either group.

Use the FIND function in a Boolean expression to identify rows that contain 
Manager in the Job_Title column.

Department        Job_Title                
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Administration    Administration Manager          
Administration    Secretary I                    
Administration    Office Assistant II            

FIND Function
The FIND function  returns the  start ing posit ion of the first  occurrence of a 
subst ring within a st ring (character value).
Find the  start ing posit ion of the subst ring Manager in the character variable 
Job_Title.

The value returned by the FIND function is 16.

Job_Title 1 2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
A d m i n i s t r a t i o n M a n a g e r

find(Job_Title,"manager","i")

FIND(string, substring<,modifier(s)><, mwsugtpos>)

Using Boolean Expressions
Part  1: Use a Boolean expression to determine whether 
an employee is a manager.

Note:  Boolean expressions evaluate to true (1) or false (0).
– If Job_Title contains Manager, the value is 1.
– If Job_Title does not  contain Manager, the value is 0 .

s103d13

proc sql;
select Department, Job_Title,       
 (find(Job_Title,"manager","i")>0)
       "Manager" 
   from mwsug.employee_information;
quit;

Viewing the Output
Part ial PROC SQL Output

Department        Job_Title                Manager
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Administration    Administration Manager         1 
Administration    Secretary I                    0
Administration    Office Assistant II            0
Administration    Office Assistant III           0
Administration    Warehouse Assistant II         0
Administration    Warehouse Assistant I          0
Administration    Warehouse Assistant III        0
Administration    Security Guard II              0
Administration    Security Guard I               0
Administration    Security Guard II              0
Administration    Security Manager               1

Using Boolean Expressions
Part  2: Calculate the stat ist ics requested.

s103d14

proc sql;
title "Manager-to-Employee Ratios";
select Department,

sum((find(Job_Title,"manager","i")>0)) as Managers,
sum((find(Job_Title,"manager","i")=0)) as Employees,
calculated Managers/calculated Employees

         "M/E Ratio" format=percent8.1
   from mwsug.employee_information
   group by Department;
quit;

Viewing the Output
PROC SQL Output

Manager-to-Employee Ratios

                                                M/E
Department              Managers  Employees    Ratio
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Accounts                       3         14   21.4%
Accounts Management            1          8   12.5%
Administration                 5         29   17.2%
Concession Management          1         10   10.0%
Engineering                    1          8   12.5%
Executives                     0          4    0.0%
Group Financials               0          3    0.0%
Group HR Management            3         15   20.0%
IS                             2         23    8.7%
Logistics Management           6          8   75.0%
Marketing                      6         14   42.9%
Purchasing                     3         15   20.0%
Sales                          0        201    0.0%
Sales Management               5          6   83.3%
Secretary of the Board         0          2    0.0%
Stock & Shipping               5         21   23.8%
Strategy                       0          2    0.0%

Handy Links

PROC SQL INTO Clause

SAS 9.4  PROC SQL User’s Guide

The Power Of SAS SQL – SAS YouTube Video

SAS Tutorial |  Step-By-Step PROC SQL – SAS YouTube Video

Working With Subquery In The SQL Procedure. Zhang, Lei. Yi, Danbo

Boolean In SQL. # ”1 Best  Programming Tip For 20 12”. Shankar, Charu

Proc Sql Syntax Order: Go Home On Time With These 5 PROC SQL Tips. Shankar, Charu

“Shankar, Charu. “Know Thy Data: Techniques For Data Explorat ion”. Pharmasug 20 18,

Ask the Expert  Webinar - Why Choose Between SAS Data Step & PROC SQL When You Can Have Both

https://go.documentation.sas.com/doc/en/pgmsascdc/v_050/sqlproc/n1tupenuhmu1j0n19d3curl9igt4.htm
https://go.documentation.sas.com/?docsetId=sqlproc&docsetTarget=titlepage.htm&docsetVersion=9.4&locale=en
https://www.youtube.com/watch?v=BOrk-qY4xjk
https://www.youtube.com/watch?v=1xyHE8qI9Hk
https://www.lexjansen.com/nesug/nesug98/dbas/p005.pdf
https://blogs.sas.com/content/sastraining/2012/05/10/1-sas-programming-tip-for-2012/
https://blogs.sas.com/content/sastraining/2012/04/24/go-home-on-time-with-these-5-proc-sql-tips/
https://www.pharmasug.org/proceedings/2018/BB/PharmaSUG-2018-BB11.pdf
https://www.sas.com/en_us/webinars/data-step-and-proc-sql.html

Charu Shankar

EMAIL Charu.shankar@sas.com 
BLOG https://blogs.sas.com/content/author/charushankar/
TWITTER CharuYogaCan
LINKEDIN https://www.linkedin.com/in/charushankar/

 Enjoy this 
session, Let  
us know in 
the evaluation

https://sas.qualtrics.com/jfe/form/SV_4MDP4cmpdnEr6V8

